Solar Radiation on Sloping Surfaces in Kentucky

By B. Barfield, J. Hill and J. Walker * Progress Report 208

UNIVERSITY OF KENTUCKY :: COLLEGE OF AGRICULTURE :: AGRICULTURAL EXPERIMENT STATION
Department of Agricultural Engineering in cooperation with National Weather Service



CONTENTS

Solar Radiation on Sloping Surfaces in Kentucky

*By B. BARFIELD, J. HILL, and J. WALKER**

The amount of solar energy (radiation flux) reaching the surface of the earth greatly affects many agricultural and industrial activities. This amount of energy is influenced by time of year as well as by the slope of the surface and the direction which the surface faces (aspect). This is known as the slope-aspect effect. A casual observation of the difference in snow cover and in the timing of tree leafing on north and south facing slopes can reveal the difference in solar energy reaching the slopes. Casual observation, however, does not indicate which slope and aspect will receive the most or the least energy for any period of the year.

Most textbooks on climatology refer to the slope-aspect effects on the quantity of solar energy reaching the surface (1). Equations are also available (2) for making calculations of the slope-aspect effect for any given date. Making use of these equations by hand computations is time-consuming; however, the use of computers makes extensive calculations practical.

The purpose of this report is to provide a set of tables and graphs showing the effect of the slope and aspect on the solar energy falling on a surface for each month of the year for the state of Kentucky.

DESCRIPTION OF THE TABLES AND GRAPHS

The amount of solar energy striking a surface at any time depends on the angle between the surface and sun and the brightness of the sun. This is shown graphically in Appendix II. If the solar intensity is known or can be assumed, calculations can be made of the amount of energy falling on any surface at any instant of time.

Climatological observations of solar energy are usually made on a horizontal surface; but in most applications involving solar energy, one is interested in the ratio of daily or monthly total energy reaching a sloping surface as compared with the horizontal surface. To determine this ratio, one has to account for the variation in angles between the surface and the sun during the day. The procedure for accounting for this variation and taking the average is shown in Appendix II. The calculated ratios are shown in Appendix I for each month of the year for Kentucky (average latitude 37.5°N).

Table 1 in Appendix I shows the average amount of solar energy falling on a horizontal surface on a clear day in Kentucky for each month. It also shows the long-term average daily radiation by months. The long-term average accounts for the effects of cloudiness. Figures 1 through 12 in Appendix I show the ratio of energy falling on a sloping surface to that on a horizontal surface for varying aspects (slope orientations) and months of the year. Figure 13 shows the effect of slope and aspect averaged over the entire year. For a given slope and aspect, one could multiply the ratio found in Fig. 1 through 13 by the appropriate value in Table 1 and determine the amount of solar energy reaching the surface on a clear day or the long-term daily average which accounts for cloudiness. For example, consider a slope of 10 degrees facing SSE (150°) during January and June. The ratio of energy reaching the slope to that on a horizontal surface is given in Figs. 1 and 6 as 1.29 in January and 1.04 in June. The long-term average reaching a horizontal surface is given in Table 1 as 610 BTU/ft^2 day in January and 2120 BUT/ft^2 day in June. Making the appropriate multiplication, the energy on the average which reaches the slope in January is 1.29 x 610 = 787 BTU/ft^2 day and in June is 1.04 x 2120 = 2205 BTU/ft^2 day.

***Associate Professor of Agricultural Engineering; Advisory Agricultural Meteorologist, National Weather Service; and Professor of Agricultural Engineering.**

The graphs in Figs. 1 through 12 are symmetrical; hence, only half of each graph is shown. To prevent "cluttering" the graphs, a portion of the curves was drawn on the left half of the graph and the remainder on the right half of the graph.

DISCUSSION OF DATA

One potential use of the data presented is in the location of plant beds. Plant beds should be located to get the maximum solar energy falling on the surface. During the months of April and May, for example, the maximum solar energy falls on the slope of 15° facing south as shown in Figs. 4 and 5. During these months, however, there is very little variation in the amount of energy received on a 15° slope varying between ESE to WSW. Hence, a plant bed would be nearly optimally located on a 15° slope facing anywhere between ESE and WSW.

Another potential use of the information is in the location of sites for exposing material to the sun to be dried. The materials to be dried could vary from lumber to digested sludge. As with the previous example, the dryer should be located and tilted so that the maximum solar radiation would be received on the surface during the drying period. In this application, one would need to consider the average expected solar energy for the months that the dryer would be operated.

A use of the data in which solar energy minimization is of interest is in the location of winter sports areas. To minimize snow melt, a slope aspect relationship should be selected which minimizes incoming solar energy.

A major potential use of the data could be for the determination of the solar intensity impinging upon windows in any type of structure. Because the data consider all slopes from horizontal to vertical, the intensity upon windows in either walls or sloped roofs can easily be determined. By appropriate use of glazing transmission data, the direct solar heating load within the structures can be determined. Possible uses of this information include the orientation of greenhouse benches and the orientation of homes and other buildings for minimum air conditioning and heating loads.

Several interesting observations can be made from the information on the graphs. One can observe for any given month that the maximum ratio will be for some angle on a south-facing slope. This angle is small in the summer months and large in the winter months. During the winter months, south-facing slopes have the largest ratio regardless of the angle. During the summer months, steep angles have a maximum ratio for SE and SW angles. This is because the sun will never be perpendicular to south-facing slopes which have steep angles.

LITERATURE CITED

1. Geiger, R. 1966. The Climate Near the Ground. Harvard University Press, Cambridge.

2. American Society of Heating, Refrigerating, and Air Conditioning Engineers. 1967. Handbook of Fundamentals. New York.

APPENDIX I:

TABLES AND GRAPHS

Table 1.—Average Solar Energy on a Horizontal Surface in Kentucky (Btu/sq ft/day).

Month	Clear Day	Long Term Average (Considering Cloudiness)
January	1,035	610
February	1,413	850
March	1,869	1,120
April	2,207	1,550
May	2,434	1,900
June	2,503	2,120
July	2,426	2,030
August	2,252	1,770
September	1,641	1,530
October	1,557	1,180
November	1,134	810
December	918	590

Table 2.—Relationship Between Slope Angles and Percent Slope.

Angle (degrees)	Slope in Percent (Ft of Fall per 100 Ft)
0	0.0
1	1.7
2	3.5
5	8.7
10	17.0
15	27.0
30	57.0
45	100.0
60	173.0
75	373.0
90	—



Fig. 1.—Slope aspect effect for January.



Fig. 2.—Slope aspect effect for February.



Fig. 3.—Slope aspect effect for March.



Fig. 4.—Slope aspect effect for April.



Fig. 5.—Slope aspect effect for May.



Fig. 6.—Slope aspect effect for June.



Fig. 7.—Slope aspect effect for July.



Fig. 8.—Slope aspect effect for August.



Fig. 9.—Slope aspect effect for September.



Fig. 10.—Slope aspect effect for October.



Fig. 11.—Slope aspect effect for November.



Fig. 12.—Slope aspect effect for December.



Fig. 13.—Average slope aspect effect over entire year.



Fig. 14.—Solar angle determinations for a sloping surface.



Fig. 15.—Solar declination.

APPENDIX II

PROCEDURE FOR CALCULATING THE RATIO OF SOLAR ENERGY FALLING ON SLOPING SURFACES

The amount of solar energy striking a surface depends on the brightness of the sun "B" and the angle between a line perpendicular (normal) to the surface and the sun. This angle "θ_s" is known as the incident angle. When the incident angle is zero, the sun is perpendicular to the surface. The amount of solar energy "Q" reaching the surface can be calculated from the equation

$$Q = B \cos\theta_s \qquad (1)$$

The incident angle can be calculated from the equation

$$\cos\theta_s = \cos\beta \cos\gamma \sin\Sigma + \sin\beta \cos\Sigma \qquad (2)$$

where β is the solar elevation angle, γ is the wall-solar azimuth, and Σ is the slope angle of the surface. These angles are shown graphically in figure 14. The wall-solar azimuth can be calculated from

$$\gamma = \psi + \psi_s \qquad (3)$$

where ψ is the wall azimuth (aspect) and ψ_s is the solar azimuth. For a given surface, the slope angle and wall azimuth will remain the same throughout the year. For given values of solar azimuth and elevation, then one can calculate the energy falling on the sloping surface with angle Σ and compare that to a horizontal surface where Σ is zero.

The solar elevation and azimuth angles can be calculated from the equation

$$\sin\beta = \sin\delta \sin L + \cos\delta \cos L \cos H \qquad (4)$$

where δ is the solar declination angle, given in Fig. 15, L is the latitude, and H is the hour angle of the sun which can be calculated from the time of day.

The hour angle of the sun can be calculated to within 4 degrees of accuracy by the equation

$$H = 15 \times \left(\begin{array}{c}\text{No. of hours from} \\ \text{noon (standard time)} \\ +\text{if a.m.} \\ -\text{if p.m.}\end{array}\right) + \left[\begin{array}{c}\text{Actual} \\ \text{Longitude}\end{array} - \begin{array}{c}\text{Reference} \\ \text{Longitude}\end{array}\right] \qquad (5)$$

The reference longitudes are 75°W, 90°W, 105°W, and 120°W for Eastern, Central, Mountain, and Pacific time, respectively. Obviously, for hour angles prior to sunrise and after sunset, H has no meaning. The hour angle can be calculated for sunrise and sunset from the equation

$$H_{sunrise} = \cos^{-1} [\tan\delta \tan L] \qquad (6)$$

A further description of Equations 1 through 6 is given in the ASHRAE Handbook of Fundamentals (2).

A computer program was developed using slightly modified forms of Equations 1 through 5 to make calculations for each hour of the day throughout each month. Calculation is made of the total daily energy falling on a sloping surface and on a horizontal surface, then the daily ratio is computed and averaged over the month. This is done for 19 different slope angles and 19 aspect angles.

The computer program for making these calculations is given in Appendix III.

The only variable is the latitude of the location being considered. The input on each card is the name of the month being computed, the Julian day number of its first day, and the total number of days in that month. The program can be used for only selected months or 12 cards can be entered to make the calculations for an entire year.

APPENDIX III:

COMPUTER PROGRAM

```
C   SOLAR
C
C
C  THIS IS A PROGRAM TO DETERMINE THE RATIO OF THE SOLAR RADIATION ON A
C  HORIZONTAL SURFACE TO THAT RECEIVED ON A SLOPING SURFACE FOR ALL
C  COMBINATIONS OF SLOPE ORIENTATION AND INCLINATION.
C
C
      IMPLICIT REAL (A-Z)
      INTEGER DAYNO, HR, DCTN,NODAYS, IMAGE,A,B,J,N,X,Y
      REAL*8 MONTH
      DIMENSION MOFCN(20,36), Z(31,20),AZANGL(31,20)
C  ESTABLISH THE LATITUDE
   10 LAT=37.5
      R=57.29578
      SINLAT=SIN(LAT/R)
      COSLAT=COS(LAT/R)
C  READ THE DATA CARD INDICATING THE NAME OF THE MONTH TO BE CALCULATED,
C  THE DAYNUMBER OF THE FIRST DAY, AND THE TOTAL NUMBER OF DAYS IN THAT
C  MONTH.
   20 READ(5,1001,END=130) MONTH, DAYNO,NODAYS
 1001 FORMAT (A8,1X,2I3)
C  COMPUTE THE POSITION OF THE SUN FOR EACH HOUR OF EACH DAY IN THE
C  MONTH AND PLACE IT IN AN ARRAY BY DAY AND HOUR.
      DO 40 N=1,NODAYS
      DO 30 HR=4,20
C  FIRST, COMPUTE THE DECLINATION OF THE SUN, DELTA
      SOLHR=12.0-HR
      HRANGL=(SOLHR*15.0)/R
      A= DAYNO+N-173
      C=.987*A
      DELTA=23.5*COS(C/R)/57.29578
C  Z IS THE ZENITH ANGLE OF THE SUN AND AZANGL IS THE AZIMUTH OF THE
C  SUN COMPUTED FROM DUE SOUTH.
      Z(N,HR)=ARCOS(SINLAT*SIN(DELTA)+COSLAT*COS(DELTA)*COS(HRANGL))
      AZANGL(N,HR)=ARSIN(-(COS(DELTA)*SIN(HRANGL)/SIN(Z(N,HR))))
   30 CONTINUE
   40 CONTINUE
C  SET THE INCLINATION OF THE SLOPE, I
      DO 100 X=1,19
      I=(5*(X-1))/R
      COSOFI=COS(I)
      SINOFI=SIN(I)
C  SET THE DIRECTION, B INITIALLY TO DUE NORTH AND CHANGE IT BY
C  INCREMENTS OF 10 DEGREES.
      DO 90 Y=1,19
      B=Y-1
C  INITIALIZE THE MONTHLY FRACTION TO ZERO BEFORE STARTING ON A
C  DIFFERENTLY ORIENTED SLOPE.
      MOTOT=0.0
C  COMPUTE THE ANGLE BETWEEN A LINE NORMAL TO THE SLOPE AND DUE SOUTH.
      APRIME=((10*B)+180)/R
      IF (APRIME .GE. 6.2832) APRIME=APRIME-6.2832
      DO 80 N=1,NODAYS
C  INITIALIZE THE TOTAL DAILY RADIATION ON THE HORIZONTAL SURFACE AND ON
```

```
C THE SLOPE TO ZERO.
      TOTHOR=0.0
      TOTSLP=0.0
      DO 70 HR=4,20
C IF THE SUN IS BELOW THE HORIZON, THERE IS NO RADIATION STRIKING THE
C SLOPE.
      IF (Z(N,HR) .GE. 1.57079) GO TO 50
C COMPUTE THE HORIZONTAL ANGLE BETWEEN THE SUN AND A NORMAL TO THE
C SLOPE.
      DIFF= AZANGL(N,HR) - APRIME
C COMPUTE THE RADIATION FALLING ON THE SLOPE ASSUMING THE INCIDENT
C RADIATION IS UNITY.
      QS=COS(Z(N,HR))*COSOFI+SIN(Z(N,HR))*SINOFI*COS(DIFF)
C WHEN QS IS NEGATIVE, THE RADIATION IS FALLING ON THE BACK OF THE
C SLOPE AND NONE IS REACHING THE FRONT.
      IF(QS .LE. 0.0) QS=0.0
C COMPUTE THE RADIATION ON A HORIZONTAL SURFACE ASSUMMING THE
C INCIDENT RADIATION IS UNITY.
      QH=COS(Z(N,HR))
      GO TO 60
   50 QS=0.0
      QH=0.0
C SUM THE HOURLY RADIATION ON THE SLOPE AND ON THE HORIZONTAL SURFACE.
   60 TOTHOR=TOTHOR+QH
      TOTSLP=TOTSLP+QS
   70 CONTINUE
C AT THE END OF THE DAY, COMPUTE THE FRACTION OF THE RADIATION
C RECEIVED ON THE SLOPE TO THAT RECEIVED ON A HORIZONTAL SURFACE.
      DAYFCN=TOTSLP/TOTHOR
C SUM THE DAILY FRACTIONS TO OBTAIN THE TOTAL FOR THE MONTH.
      MOTOT=DAYFCN+MOTOT
   80 CONTINUE
C DIVIDE THE MONTHLY TOTAL BY THE NUMBER OF DAYS IN THE MONTH IN ORDER
C TO OBTAIN THE AVERAGE MONTHLY RATIO OF RADIATION ON THE SLOPE TO
C RADIATION ON A  HORIZONTAL SURFACE FOR THE GIVEN SLOPE.
      MOFCN(X,Y)=MOTOT/NODAYS
C INCREMENT THE INCLINATION AND ORIENTATION OF THE SLOPE TO OBTAIN
C ALL POSSIBLE COMBINATIONS.
   90 CONTINUE
  100 CONTINUE
  110 WRITE (6,1002 ) MONTH
 1002 FORMAT ('1',42X,'DURING ',A8,' THE FRACTION OF THE SOLAR RADIATION
     X ON'/,43X,'A HORIZONTAL SURFACE WHICH IS RECEIVED ON A SLOPE IS:')
      WRITE (6,1003)
 1003 FORMAT ('0',47X,'SLOPE ANGLE IN DEGREES (0 IS LEVEL GROUND)')
      WRITE(6,1004) (J, J=5,90,5)
 1004 FORMAT ('0',22X,'0',18I5)
      DO 120 Y=1,19
      B=Y-1
      DCTN=10*B
      IMAGE=360-DCTN
      WRITE(6,1005) DCTN, (MOFCN(X,Y), X=1,19), IMAGE
 1005 FORMAT (14X,I3,3X,19F5.2,4X,I3)
  120 CONTINUE
      GO TO 20
  130 STOP
      END
```

2M—3-73